Main menu

Tag Archives: user interface

lesson 32 started looking at how to export individual pages as separate PDFs. it looks like this topic has become popular … a fellow aussie with a passion for getting the most out of InDesign, with a particular focus on prepress, has also written about this, outlining a few solutions — Breaking up is hard to do…

as colmin8r points out, the script from lesson 32 has no user interface, so you can’t control how files are named. that’s what this lesson is all about — extending the script to give you more options. this is how scripts develop — get the basic functionality working, then expand the script to make it more user-friendly, or adaptable, or whatever.

here’s our base script — this time we’re exporting an interactive PDF, rather than using a print PDF preset — but apart from that, it’s the same as the script from lesson 32 :

tell application id "com.adobe.InDesign"
set raster resolution of interactive PDF export preferences to one hundred forty four ppi
set (export reader spreads of interactive PDF export preferences) to false
set mgFolder to file path of active document
set mgDocName to name of active document
end tell
tell application "Finder"
if (exists folder "the pdfs" of folder mgFolder) is false then
make folder at mgFolder with properties {name:"the pdfs"}
end if
end tell
tell application id "com.adobe.InDesign"
-- REST OF FUNCTIONALITY WILL GO IN HERE
repeat with x from 1 to count pages of active document
set mgPageName to name of page x of active document
set page range of interactive PDF export preferences to mgPageName
set mgFilePath to mgFolder & "the pdfs:" & mgDocName & mgPageName & ".pdf" as string
tell active document
export format interactive PDF to mgFilePath without showing options
end tell
end repeat
display dialog "ALL DONE!" buttons {"OK"} default button 1
end tell

we’re going to give the user two options — shortening the existing filename by breaking it at the first space or underscore, and the option to add something else to the filename. the code for the dialog looks like this :

set mgDialog to make dialog
tell mgDialog
tell (make dialog column)
tell (make dialog row)
make static text with properties {static label:"Where do you want to break the filename?"}
end tell
tell (make dialog row)
set mgDelimiterButtons to make radiobutton group
tell mgDelimiterButtons
make radiobutton control with properties {static label:"first underscore", checked state:true}
make radiobutton control with properties {static label:"first space"}
make radiobutton control with properties {static label:"full stop", min width:290}
end tell
end tell
tell (make dialog row)
make static text with properties {static label:"current filename : " & mgDocName}
end tell
tell (make dialog row)
make static text with properties {static label:" "}
end tell
tell (make dialog row)
make static text with properties {static label:"Enter the text you'd like to append."}
end tell
tell (make dialog row)
set mgSuffixField to make text editbox with properties {edit contents:"(can be blank)", min width:250}
end tell
end tell
end tell
set mgResult to show mgDialog

… and creates a user interface that looks like this :
a couple of things to notice … we’ve also added the option of breaking the filename at the first full stop which, if you’re naming your files correctly, means not shortening the filename at all. the dialog also shows the existing filename (in this case “export separate pages_test_file.indd”) to make it easier for the user to make their selections.

the remainder of the code takes the information from the user interface and constructs a new filename to use when saving the PDFs :

if mgResult is true then
set mgDelimiterButton to selected button of mgDelimiterButtons
if mgDelimiterButton is 0 then
set mgDelimiter to "_"
else if mgDelimiterButton is 1 then
set mgDelimiter to " "
else if mgDelimiterButton is 2 then
set mgDelimiter to "."
end if
set text item delimiters of AppleScript to mgDelimiter
set mgDocName to text item 1 of mgDocName
set text item delimiters of AppleScript to ""
set mgSuffix to edit contents of mgSuffixField as string
if mgSuffix is "(can be blank)" then
set mgSuffix to ""
end if
set mgDocName to (mgDocName & "_" & mgSuffix)
destroy mgDialog
else
error number -128
end if

two things to note here. we’ve added some error handling just in case the user inadvertently leaves “(can be blank)” in the text field of the user interface — we don’t want that to become part of the filename, so we change it to “” (that is, nothing) instead. also, we’re using an underscore to join the shortened filename and the suffix.

hopefully it’s pretty clear how this filenaming works. as an example — if we start with the filename “export separate pages_test_file.indd”, break it at the first space and add the suffix “concepts”, our resulting PDF filenames will be “export_concepts1.pdf”, “export_concepts2.pdf”, etc.

ok, that’s one way you could do it — but you could also make it more complex again by giving the user the option to break the filename in a different place. of course, more complex options mean more complex code. so this :

set mgDialog to make dialog with properties {name:"Where do you want to break the filename?"}
tell mgDialog
tell (make dialog column)
set mgPlaceButtons to make radiobutton group
tell mgPlaceButtons
make radiobutton control with properties {static label:"first", checked state:true}
make radiobutton control with properties {static label:"second"}
make radiobutton control with properties {static label:"third", min width:100}
end tell
end tell
tell (make dialog column)
set mgDelimiterButtons to make radiobutton group
tell mgDelimiterButtons
make radiobutton control with properties {static label:"space", checked state:true}
make radiobutton control with properties {static label:"underscore"}
make radiobutton control with properties {static label:"full stop", min width:130}
end tell
end tell
tell (make dialog column)
tell (make dialog row)
make static text with properties {static label:"current filename : "}
end tell
tell (make dialog row)
make static text with properties {static label:mgDocName}
end tell
tell (make dialog row)
make static text with properties {static label:" ", min width:300}
end tell
tell (make dialog row)
make static text with properties {static label:"Enter the text you'd like to append."}
end tell
tell (make dialog row)
set mgSuffixField to make text editbox with properties {edit contents:"(can be blank)", min width:250}
end tell
end tell
end tell
set mgResult to show mgDialog

is what you need to create a user interface that looks like this :

and the rest of the code is more complex too. this time we’re using the chosen break character to join the filename to the suffix (rather than forcing it to be an underscore) :

if mgResult is true then
set mgPlaceButton to selected button of mgPlaceButtons
if mgPlaceButton is 0 then
set mgPlace to "1"
else if mgPlaceButton is 1 then
set mgPlace to "2"
else if mgPlaceButton is 2 then
set mgPlace to "3"
end if
set mgDelimiterButton to selected button of mgDelimiterButtons
if mgDelimiterButton is 0 then
set mgDelimiter to " "
else if mgDelimiterButton is 1 then
set mgDelimiter to "_"
else if mgDelimiterButton is 2 then
set mgDelimiter to "."
end if
set text item delimiters of AppleScript to mgDelimiter
try
set mgDocName to text items 1 thru (mgPlace + 1) of mgDocName
set mgDocName to text items 1 thru mgPlace of mgDocName
on error
display dialog "come on dude — that's not possible" & return & "there are not " & mgPlace & " \"" & mgDelimiter & "\" in the filename" & return & return & "have another go" buttons "sorry"
error number -128
end try
set mgDocName to mgDocName as string
set text item delimiters of AppleScript to ""
set mgSuffix to edit contents of mgSuffixField as string
if mgSuffix is "(can be blank)" or mgSuffix is "" then
set mgSuffix to ""
set mgDocName to (mgDocName & mgSuffix)
else
set mgDocName to (mgDocName & mgDelimiter & mgSuffix)
end if
destroy mgDialog
else
error number -128
end if

so this time — if we start with the filename “export separate pages_test_file.indd”, break it at the second space and add the suffix “concepts”, our resulting PDF filenames will be “export separate concepts1.pdf”, “export separate concepts2.pdf”, etc.

notice in the above code we’ve added some error handling to make sure the user’s choices make sense. that is, if they choose to break the filename at the third space, and there are not three spaces in the filename, rather than the script simply failing, they’ll get a message stating why it’s not going to work :

and, of course, you can keep on expanding the script until you have as many different options as suits your needs.

the next lesson will look at a couple of other ways we could approach the naming of these separate page PDFs — automatically generating filenames based on data already existing in the document.